
Как рассчитать число вхождений текста A в текст B (как часто 

некоторое слово встречается в ячейке) 

Файл с примером 

В Google Таблицах (как и в Excel) нет встроенной функции для решения такой задачи - выяснения, сколько раз 

какое-то определенное слово встречается в тексте (в ячейках). 

А пригодится она может в разных ситуациях: например, при анализе поисковых запросов (сколько раз 

определенное слово повторяется в том или ином запросе?). Или - представим - вам нужно проанализировать, 

сколько раз в статьях вашего корпоративного блога встречается название вашей компании (кстати - тексты 

статей из блога можно загружать прямо в Таблицы с помощью функции IMPORTFEED, она несложная и мы не 

будем здесь на ней отдельно останавливаться). 

Как и со многими другими задачами, ее можно решить тем или иным способом с помощью комбинирования 

нескольких функций. 

В нашем примере в столбце A есть несколько текстовых строк, и наша задача — выяснить, сколько раз в каждой 

из них встречается искомый текст, расположенный в ячейке A1 (голубого цвета): 

 
Рис. В столбце B вручную указаны ответы, которые мы хотим получать с помощью формулы автоматически. 

Для решения этой задачи можно воспользоваться сложной формулой, состоящей из следующих функций: 

1.   ДЛСТР / LEN, которая, напомню, вычисляет длину текста в символах.  

2.   ПОДСТАВИТЬ / SUBSTITUTE — которая заменяет в текстовой строке определённый текст другим. 

Чтобы найти вхождение определенной текстовой строки в другую, нужно удалить все её вхождения в исходную 

и сравнить длину полученной строки с исходной: 

Например, в данном случае нам нужно узнать, сколько раз встречается слово “МТС”  - удалим его из текста и 

сравним, сколько символов исчезло. 

ДЛСТР(“Тариф МТС Супер МТС”) – ДЛСТР(“Тариф Супер”) = 6 

Исчезло 6 символов. Разделим это число на длину того слова, которое мы “подсчитывали” (это 3 буквы):  

6 / ДЛСТР (“МТС”) = 2 

Получается, что два раза слово «МТС» (состоящее из трех символов) входит в исходную ячейку с текстом.  

Осталось записать этот алгоритм на языке формул. 

Будем удалять искомый текст из того, где мы его ищем, с помощью ПОДСТАВИТЬ: 

https://docs.google.com/spreadsheets/d/1Ox0aLx0GneaNnov7oBxz3tifVzuhTnIe4YtMljassFA/edit?usp=sharing


=ПОДСТАВИТЬ(где ищем;что ищем;"") 

А затем вычислять разницу между исходным текстом и текстом после удаления искомого слова (если оно 
встречается). Эта формула будет вычислять разницу - сколько символов исчезло после удаления искомого 

слова в тексте: 

=(ДЛСТР(где ищем)-ДЛСТР(ПОДСТАВИТЬ(где ищем;что ищем;""))) 

Наконец, разделим полученное число на длину нашего искомого слова, чтобы выяснить, сколько раз оно  

встречалось в искомом тексте: 

=(ДЛСТР(где ищем)-ДЛСТР(ПОДСТАВИТЬ(где ищем;что ищем;"")))/ДЛСТР(что ищем) 

В нашем примере формула выглядит следующим образом: 

=(ДЛСТР(A2)-ДЛСТР(ПОДСТАВИТЬ(A2;$A$1;"")))/ДЛСТР($A$1) 

 
Рис. Итоговый вид формулы, которая отвечает на вопрос: сколько раз слово из ячейки A1 встречается в тексте?  

